Limites trigonométriques remarquables

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

Nous allons illustrer et démontrer dans ce document deux limites importantes dans l'étude du calcul différentiel.

$$\lim_{\theta \to 0} \frac{\sin(\theta)}{\theta} = 1 \text{ et } \lim_{\theta \to 0} \frac{1-\cos(\theta)}{\theta} = 0.$$

Chacune de deux limites est une forme indéterminée du type $\frac{0}{0}$. Ces indéterminations seront levées de manière géométrique pour la limite $\lim_{\theta \to 0} \frac{\sin(\theta)}{\theta}$ et de manière algébrique pour la limite $\lim_{\theta \to 0} \frac{1-\cos(\theta)}{\theta}$.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation

```
> restart;
  with(plots,display,implicitplot,pointplot,textplot):
  with(plottools,sector):
> Fond:=ColorTools:-Color([1,0.99215686,0.96862745]);
```

$$Fond := \langle \text{RGB} : 1\ 0.992\ 0.969 \rangle \qquad \textbf{(1.1)}$$

La procédure suivante transpose graphiquement le théorème du sandwitch.

Procédure Sandwitch

```
> Sandwitch := proc(angle)
    local A,C,Opts,P,S,Secteur,theta,T,Title,t;
    Opts:=labels=[``,``],titlefont=[times,italic,14],axis=[color="Niagara
  1"]:
    C := implicitplot( x^2+y^2=1, x=-1..1, y=-1..1, tickmarks=[0,0],color =
  "Niagara 1",thickness=0):
    S := angle -> plot( [[0,0],[cos(angle),sin(angle)],[cos(angle),0]],
  color="Resene CeSoir" ):
    A := angle -> plot( [cos(angle)*cos(t),cos(angle)*sin(t),t=0..angle],
  color="Resene GrannySmith" ):
    Secteur:=angle->sector([0, 0],cos(angle),0..angle,color="Resene
  GrannySmith");
    P := angle -> pointplot([[1,0],
                            [cos(angle),0],
                            [cos(angle)*cos(angle),cos(angle)*sin(angle)],
```

```
                            [cos(angle),sin(angle)] ],
                            symbol=solidcircle, symbolsize=11,color=
"Niagara 1"):
  if evalf(angle)>0 then T:=angle -> textplot({
                  [-0.085,-0.09,typeset("O")],
                  [1.20,-0.1,"A (1,0)"],
                  [cos(angle),-0.1,"B"],
                  [cos(angle)*cos(angle)-0.1,cos(angle)*sin(angle)+0.1,
"C"],
                  [cos(angle)+0.40,sin(angle)+0.1,typeset("P (cos ",
theta, ", sin ",theta,")")],
                  [-0.15,1.1,"(0,1)"],
                  [0.15,0.06,typeset(theta)]}):
  else T:=angle -> textplot({
                  [-0.085,-0.09,"O"],
                  [1.20,0.1,"A (1,0)"],
                  [cos(angle),0.1,"B"],
                  [cos(angle)*cos(angle)-0.05,cos(angle)*sin(angle)-0.1,
"C"],
                  [cos(angle)+0.47,sin(angle)+0.025,typeset("P (cos ",
theta, ", sin ",theta,")")],
                  [-0.15,1.1,"(0,1)"],
                  [0.15,-0.06,typeset(theta)]}):
  fi:
  Title:=proc(angle)
      if nargs > 1
        then
          if args[2]=Base
           then typeset("Cercle unité où ",theta," = ",sprintf( "%10.8f",
angle))
          elif args[2]=Complet
           then typeset("Lorsque ",theta," = ",sprintf("%10.8f",angle),
                 "\n Aire du secteur circulaire COB = ",sprintf("%10.8f",
abs(angle)*cos(angle)^2/2),
                 "\n Aire du triangle rectangle POB = ",sprintf("%10.8f",
cos(angle)*abs(sin(angle))/2),
                 "\n Aire du secteur circulaire POA = ",sprintf("%10.8f",
abs(angle)/2))
          end if
         else return ""
      end if
    end proc;
  plots[display]( [C,S(angle),A(angle),P(angle),Secteur(angle),T(angle)],
background=Fond,
                  view=[-1.5..1.5,-1.5..1.5],
                  title=Title(args),
                  scaling=constrained,Opts):
end proc:
```

Énoncé du théorème du sandwich

Si $\lim_{x\to a} f(x) = L = \lim_{x\to a} g(x)$ et si $\forall x \in V_o(a)$, $f(x) \le h(x) \le g(x)$ alors $\lim_{x\to a} h(x) = L$.

Le théorème du sandwich permet d'établir indirectement la limite d'une fonction. Si deux fonctions (f et h) admettent la même limite en $x = a$ et qu'une troisième fonction (g) est encadrée ou « prise en sandwich » entre f et h dans un voisinage troué de a, alors g admet en $x = a$ une limite et celle limite est égale à la limite commune de f et h.

Preuve de $\lim_{\theta \to 0} \frac{\sin(\theta)}{\theta} = 1$

Nous allons d'abord considérer quelques illustrations à l'aide de la procédure Sandwitch.

D'abord avec $\theta > 0$:

```
> Sandwitch1:=Sandwitch(Pi/4,Base):
  Sandwitch2:=Sandwitch(Pi/8,Base):
  Sandwitch3:=Sandwitch(Pi/16,Base):
  display(Matrix(1,3,[Sandwitch||(1..3)]));
```



Dans le premier quadrant ($\theta > 0$), chaque tracé met en évidence trois régions géométriques avec $\theta = \frac{\pi}{4}$, $\theta = \frac{\pi}{8}$ et $\theta = \frac{\pi}{16}$;

- Secteur circulaire COB
- Triangle rectangle POB
- Secteur circulaire POA

Ensuite avec $\theta < 0$:

```
> Sandwitch4:=Sandwitch(-Pi/4,Base):
  Sandwitch5:=Sandwitch(-Pi/8,Base):
  Sandwitch6:=Sandwitch(-Pi/16,Base):
```

```
display(Matrix(1,3,[Sandwitch||(4..6)]));
```



Dans le quatrième quadrant ($\theta < 0$), chaque tracé met en évidence trois régions géométriques avec $\theta = -\frac{\pi}{4}$, $\theta = -\frac{\pi}{8}$ et $\theta = -\frac{\pi}{16}$;

- Secteur circulaire COB
- Triangle rectangle POB
- Secteur circulaire POA

L'idée derrière ces illustrations est de vous faire visualiser que lorsque $\theta \to 0^+$ aussi bien que lorsque $\theta \to 0^-$, nous avons toujours

Aire du secteur circulaire COB $\leq$ Aire du triangle rectangle POB $\leq$ Aire du secteur circulaire POA

Illustrons-le également avec une petite animation. Lorsque les tracés seront obtenus, sélectionner l'un des deux tracés puis, dans la barre de menu contextuel, cliquer sur le bouton Démarrer/Poursuivre l'animation pour lancer l'animation.

```
> ThetaI:=[seq(Pi/60*(31-i),i=1..30)]:
  ThetaIV:=[seq(-Pi/60*(31-i),i=1..30)]:
  FramesI:=[seq(Sandwitch(t,Complet),t=ThetaI)]:
  FramesIV:=[seq(Sandwitch(t,Complet),t=ThetaIV)]:
  A:=display(FramesI,insequence=true):
  B:=display(FramesIV,insequence=true):
  display( Matrix(1,2,[A,B]),background=Fond);
```

Lorsque θ = 1.57079633 *Aire du secteur circulaire COB = 0.00000000* *Aire du triangle rectangle POB = 0.00000000* *Aire du secteur circulaire POA = 0.78539816*  	*Lorsque θ = -1.57079633* *Aire du secteur circulaire COB = 0.00000000* *Aire du triangle rectangle POB = 0.00000000* *Aire du secteur circulaire POA = 0.78539816* 

Chaque titre des animations précédentes affiche les résultats des calculs d'aires de chaque région avec trente valeurs de $\theta \to 0$ et nous avons toujours les inégalités suivantes vérifiées.

Aire du secteur circulaire COB $\leq$ Aire du triangle rectangle POB $\leq$ Aire du secteur circulaire POA

De plus, lorsque $\theta \to 0$, il semble que la valeur de l'aire de ces trois régions tendent à être égales.

Cette approche géométrique est utile à la démonstartion de l'existence de la limite $\lim_{\theta \to 0} \frac{\sin(\theta)}{\theta}$.

La preuve rigoureuse de cette limite passera par des théorèmes sur les limites, en particulier celui du théorème du sandwitch.

Rappelons les formules du calcul de l'aire de ces régions:

 Aire d'un secteur circulaire d'angle θ (radians) d'un disque de rayon r est donnée par la formule $A = \frac{1}{2} r^2 \theta$	 	Aire du secteur circulaire COB $= \frac{1}{2} \overline{OB}^2 \theta = \frac{\lvert\theta\rvert}{2} \cos(\theta)^2$ Aire du triangle rectangle POB $=$ $\frac{base\, x\, hauteur}{2} = \frac{\cos(\theta) \lvert\sin(\theta)\rvert}{2}$ Aire du secteur circulaire POA $= \frac{1}{2} \overline{OA}^2 \theta = \frac{\lvert\theta\rvert}{2}$

Nous avons donc

Aire du secteur circulaire COB $\leq$ Aire du triangle rectangle POB $\leq$ Aire du secteur circulaire POA

$$\Leftrightarrow$$

$$\frac{\lvert\theta\rvert}{2} \cos(\theta)^2 \leq \frac{1}{2} \cos(\theta) \lvert\sin(\theta)\rvert \leq \frac{\lvert\theta\rvert}{2}$$

Considérons $\theta \neq 0$ (puisque nous allons calculer une limite où $\theta \to 0$) et puisque $\cos(\theta) \neq 0$ (θ est soit dans premier quadrant ou dans le quatrième). Multiplions les membres de ces inégalités par $\dfrac{2}{|\theta| \cos(\theta)} > 0$. Nous obtenons

$$\cos(\theta) \le \frac{|\sin(\theta)|}{|\theta|} \le \frac{1}{\cos(\theta)}$$

$\forall\, \theta \in \left]-\frac{\pi}{2}, \frac{\pi}{2}\right[$ $(\theta \neq 0)$, $\dfrac{\sin(\theta)}{\theta} > 0$ et ainsi $\dfrac{|\sin(\theta)|}{|\theta|} = \dfrac{\sin(\theta)}{\theta}$. Nous pouvons alors écrire

$$\cos(\theta) \le \frac{\sin(\theta)}{\theta} \le \frac{1}{\cos(\theta)}$$

$$\Leftrightarrow$$

$$\lim_{\theta \to 0} \cos(\theta) \lim_{\theta \to 0} \frac{\sin(\theta)}{\theta} \le \lim_{\theta \to 0} \frac{1}{\cos(\theta)}$$

$$\Leftrightarrow$$

$$1 \le \lim_{\theta \to 0} \frac{\sin(\theta)}{\theta} \le 1$$

Visualisons l'encadrement de $\dfrac{\sin(\theta)}{\theta}$ par $\cos(\theta)$ et $\dfrac{1}{\cos(\theta)}$.

```
> plot([1/cos(theta),sin(theta)/theta,cos(theta)],
       theta=-1..1,color=["Resene 186","Resene 168","Resene 216"],
       thickness=[3,1,3],
       titlefont=[times,italic,14],
       title=typeset("Théorème du sandwitch pour ", sin(theta)/theta,"
 au voisinage de ",theta=0),
       legend=[typeset(y=1/cos(theta)),typeset(y=sin(theta)/theta),
 typeset(y=cos(theta))],
       legendstyle=[location=top]);
```



Le théorème du sandwich nous permet de conclure que $\lim_{\theta \to 0} \frac{\sin(\theta)}{\theta} = 1$.

Preuve de $\lim_{\theta \to 0} \frac{1 - \cos(\theta)}{\theta} = 0$

La preuve que nous allons faire n'est pas géométrique. La preuve sera faite en utilisant les théorèmes sur les limites et le résultat prouvé précédemment : $\lim_{\theta \to 0} \frac{\sin(\theta)}{\theta} = 1$.

Commençons par transformer $\frac{1 - \cos(\theta)}{\theta}$ en multipliant cette expression par 1 où 1 sera exprimé par le quotient $\frac{1 + \cos(\theta)}{1 + \cos(\theta)}$ en considérant que $\theta \to 0$.

$$\begin{aligned} \frac{1 - \cos(\theta)}{\theta} &= \frac{1 - \cos(\theta)}{\theta} \frac{1 + \cos(\theta)}{1 + \cos(\theta)} \\ &= \frac{1 - \cos(\theta)^2}{\theta\,(1 + \cos(\theta))} \\ &= \frac{\sin(\theta)^2}{\theta\,(1 + \cos(\theta))} \\ &= \frac{\sin(\theta)}{\theta} \frac{\sin(\theta)}{1 + \cos(\theta)} \end{aligned}$$

Il s'en suit que

$$\begin{aligned} \lim_{\theta \to 0} \frac{1 - \cos(\theta)}{\theta} &= \lim_{\theta \to 0} \left(\frac{\sin(\theta)}{\theta} \cdot \frac{\sin(\theta)}{1 + \cos(\theta)} \right) \\ &= \left(\lim_{\theta \to 0} \frac{\sin(\theta)}{\theta} \right) \left(\lim_{\theta \to 0} \frac{\sin(\theta)}{1 + \cos(\theta)} \right) \\ &= (1).(0) \\ &= 0 \end{aligned}$$

Bingo !